
ABSTRACT
The 2016 Monitoring the Future survey is part of an annual, long-term study of American adolescents and
adult high school graduates conducted by the University of Michigan’s Institute for Social Research. This
secondary data analysis uses the FACTOR procedure in the SASTM Studio software to perform factor
analysis to extract latent structures describing academic achievement, environment, and student
delinquency. 17,719 observations were used to perform multivariate analysis of variance via the GLM
procedure to explore the relationships between the extracted factors and demographic variables for
ethnicity, gender, and population density. The SAS Code and results are presented, along with a
discussion of the necessary data cleaning steps, data quality assessment and post-hoc analyses. A
statistically significant overall effect was found for each independent variable: gender (Wilks’ Lambda =
0.96, F= 273.23, df= (3, 17711), p < 0.0001); race (Wilks’ Lambda = 0.86, F= 479.83, df= (6, 35422), p <
0.0001); population density (Wilks’ Lambda = 0.98, F= 65.31, df= (6, 35422), p < 0.0001). Population
density explains 2% (Pillai’s trace = 0.022, p < 0.0001) of the variance in academic achievement,
academic environment, and at-risk behaviors. Gender explains 4% (Pillai’s trace = 0.044, p < 0.0001),
and race explains 14% (Pillai’s trace = 0.145, p < 0.0001) of the variance. The academic environment for
8th and 10th grade students was described by an extracted factor with high loadings for the variables for
parental education, college preparatory program, and remedial schooling (negative loading), and was
shown to vary significantly by race.

INTRODUCTION
Previous research has shown that urbanicity is an important indicator for understanding adolescent
academic achievement (Miller & Vortruba-Drzal, 2015) and that gender differences in achievement
favoring female students persist across subject domains (Voyer, Voyer, & Hinshaw, 2014). Additionally,
gender differences remain significant after controlling for socio-economic background (Carvalho, 2016).
This study uses SASTM1 Studio software and a publicly available dataset to explore the relationships
between demographic variables for race, gender, and population density and extracted factors describing
academic achievement, academic environment and at-risk behaviors. The FACTOR procedure will be
used to perform factor analysis and extraction. The GLM procedure will be used to perform multivariate
analysis of variance (MANOVA).
The 2016 Monitoring the Future (MTF) 8th- and 10th-Grade surveys are part of an annual, long-term study
of American adolescents and adult high school graduates conducted by the University of Michigan’s
Institute for Social Research. Study data have been collected since 1975, when the Institute of Social
Research opened. The 2016 MTF survey involved 45,500 participants in grades 8, 10 and 12, and
sampled from 372 secondary schools across the country (Johnston, O'Malley, Miech, Bachman, &
Schulenberg, 2017). With an estimated 500 variables per year, the survey covers a broad range of topics
including drug use, attitudes and beliefs regarding drug use, and lifestyle choices and values (Johnston,
Bachman, O'Malley, Schulenberg, & Miech, Codebook., 2016). The data and codebook are publicly
available via the Inter-University Consortium for Political and Social Research (ICPSR) website
(Johnston, Bachman, O'Malley, Schulenberg, & Miech, 2017). Factor analysis can be used to
mathematically define underlying data structures which reflect characteristics that cannot or were not
directly queried. This secondary data analysis will use those techniques to extract the factors for student
achievement, their academic attitudes and environment, and at-risk behaviors or delinquency. The
extracted factors will then be used as the dependent variables in multivariate analysis of variance
(MANOVA). The MANOVA analysis will evaluate the relationship between the demographic independent

variables for gender, ethnicity, and population density and the dependent factors for academic success,
academic environment, and at-risk behaviors. Race, gender, and urbanicity are hypothesized to have
significant relationships with academic achievement and delinquency, and gender is not expected to be
statistically significantly related to academic environment.

DATA QUALITY
The 2016 MTF dataset for 8th- and 10th-Grade survey responses has 32,873 responses for 571
variables. Four survey forms were used for data collection at the 8th and 10th grade levels and not all
questions were asked on all forms (Johnston, Bachman, O'Malley, Schulenberg, & Miech, Codebook.,
2016). Prior exploratory analysis of the 2016 MTF dataset indicated that underlying factors could be
identified and extracted to represent student academic achievement, attitudes toward education, and
delinquent or at-risk behaviors. A subset of 14 variables from the exploratory analysis were chosen for
factor analysis and were hypothesized to represent these three major groupings. To increase the number
of complete cases for factor analysis, the selected variables were chosen from those that appeared on all
four survey forms, where possible. Demographic variables for grade level, gender, ethnicity, and
metropolitan statistical area (MSA) were retained to facilitate the MANOVA analysis and investigations
into missing data. Analysis was performed in SASTM Studio, release 3.7 (Enterprise Edition, Build Jun 11,
2018). The supporting code is available in Appendix A.

DATA QUALITY EVALUATION
The MEANS procedure was used to investigate missing data and extreme values in the 14 variables
selected for factor analysis. The results are presented in Table 1. Less than 7.5% of the data were
missing for each variable. Variables except respondent’s high school program type (V7222) were either
dichotomous (0-1, or 1-2), rated on a Likert scale (1-4, or 1-5), ordinal, or discretized continuous variables
represented on an ascending scale. The variable for high school program type consisted of four nominal
levels for high school program with no implicit ranking beyond “college prep” vs “all others.” No variables
had scores out of range or extremely low variance. The variables for high school program type (V7222),
father’s education level (V7215) and mother’s education level (V7216) will be re-coded in the following
sections to ease interpretation of the factor analysis.

Table 1. Descriptive Statistics for Numeric Variables before Recoding or Parceling
Variables V508 and V509 are component variables that together specify a standardized 3-category
measure of population density, indicating whether the associated region is a large or medium
Metropolitan Statistical Area (MSA) or fails to qualify for the MSA designation and has been scored “nonMSA.” These variables were additively combined to produce a measure for population density,
“pop_density,” as described in the 2016 MTF Codebook (Johnston, Bachman, O'Malley, Schulenberg, &
Miech, Codebook., 2016). The SUM function in SAS was not used, as assigning a zero value to missing
data for either variable would yield a misleading population density designation.

The ‘+’ operator was used so that a missing value for either V508 or V509 would result in a missing value
for pop_density. The FREQ procedure was used to investigate missing data and proportions of the
categorical demographic variables retained for MANOVA analysis and the student grade level (V501).
The proportions of male and female respondents were relatively similar. 22% of observations were
missing data for ethnicity. Over half (57.2%) of the respondents who reported their race were White
(V1070=2). Almost half (47.1%) of respondents were from moderately-populated areas (pop_density=1).

Figure 1. Frequencies and Proportions for Categorical Demographic Variables
The UNIVARIATE procedure was used to examine the distributions of the numeric variables. A highly
right-skewed distribution was observed for V7102 (CS = 9.08), such that the 95th percentile of all
observations was 1, the lowest value in a 7-point scale for the number of cigarettes smoked in the past 30
days. A response of 1 corresponds with no cigarette use at all. This right-skewed distribution is consistent
with findings of declining smoking initiation and cigarette use in adolescents by the MTF principal
investigators (Johnston, O'Malley, Miech, Bachman, & Schulenberg, 2017).

DATA CLEANING AND PARCEL CREATION
Variables for high school program type (V7222), father’s education level (V7215), mother’s education
level (V7216), repeating a grade (V7232), and summer school (V7233) were recoded and/or parceled
prior to factor analysis. Variable V7222 consisted of four nominal levels for high school program and was
recoded as a dichotomous variable for whether the student was in a college preparatory program (0=no,
1=yes). For parental education levels, V7215 and V7216, the “don’t know” category was originally coded
as level 7 of an ordered scale of ascending levels of education (1= grade school, 6= grad school). These
values for “don’t know” were recoded as missing, and V7215 and V2716 were additively combined to a
parcel for parent education (parent_ed). A parcel for remedial schooling (rem_school) additively combines
the dichotomous variables for whether a respondent has ever needed summer school (V7233) or had
ever been held back a grade (V7232). V7232 and V7233 showed low variance in the initial data quality

assessment (Table 1). These recoding and parceling steps are consistent with the prior analysis that
identified the underlying factors of interest.

MISSING DATA
To assess the impact of missing data on the number of complete cases for analysis, a dummy variable
“anymiss” was created which was coded 1 when the respondent had one or greater missing values
across all variables and 0 when the case was complete across the newly created parcels, the numeric
variables for factor analysis, and the population density demographic variable. 17,719 (53.9%)
observations were complete for these 15 variables. A second dummy variable, “missdata,” was created to
reflect the count of the number of missing values for each record. 95% of records were missing values for
4 or fewer variables. The PROC FREQ output for these variables is presented in Appendix B, Figure 7.
Cross-tabulation tables of anymiss and grade level, gender, ethnicity were populated, and Pearson’s chisquare test for association was performed for each comparison. These tables and the output of the
statistical tests are presented in Appendix B, Figure 8 - Figure 10. Statistically significant associations
were found between anymiss and all demographic variables investigated, while the proportion of
complete cases was within ± <5% of the observed proportion of complete cases in the entire sample, for
all comparisons except ethnicity. A t-test was performed to evaluate the differences in the number of
missing values (missdata) by grade level (V501). Though the t-test (unequal variances) demonstrates a
statistically significant difference in means for grade 8 and grade 10 (t = 13.30, df = 32,862, p <0.0001),
the observed difference in means is less than one missing item (0.2842) and is not practically significant.
Figure 2 shows the histograms for missdata by grade level. Both distributions are highly right-skewed and
span a similar range of responses. See Appendix B, Figure 11 for the TTEST procedure’s statistical
output.

Figure 2. Distributions of Count of Missing Values by Grade

FACTOR ANALYSIS
Factor analysis was performed in SASTM Studio using the FACTOR procedure. The REORDER option
was used to sort variables by their factor loadings and the SCREE option was used to produce a scree
plot. Computation of the parallel analysis criterion for factor retention was performed using a script
previously published by Brian O’Connor (2000). The parallel analysis criterion offers an alternative to the

Kaiser criterion, which may retain too many factors, and visual inspection of the scree plot, which may be
considered too subjective. The eigenvalue of a factor should exceed the 95th percentile eigenvalue of
the similarly-numbered factor as generated by the parallel analysis criterion script. The output from this
script is presented in Figure 12, Appendix B. Principal components analysis with a communality estimate
of one was used for extracting factors. The Kaiser criterion, parallel analysis criterion, and scree plot
shown in Figure 3 each suggest three factors, as expected. For 22,079 observations, the 95th percentile
of eigenvalues for the fourth factor from randomly generated data was 1.020. The eigenvalue for factor 4
in this analysis was 0.977. The table of Eigenvalues is given in Figure 13, Appendix B.

Figure 3. Scree Plot from Factor Analysis
Figure 4 shows the resulting rotated factor pattern, after varimax rotation. The first factor, representing
academic achievement and attitudes, had high loadings (>0.4) for “I often do my best work,” high school
grades, and the student’s projections for whether he or she would graduate and whether he or she would
attend a 4-year college. This factor loaded negatively on “I often fail to complete or turn in my
assignments.” The second factor, representing academic environment, had high loadings for parental
education and college preparatory program, and negative loading for remedial schooling interventions.
The last factor, representing delinquent or at-risk behaviors, had high loadings for drinking, alcohol use,
friends who had dropped out, and skipping class.

Figure 4. Rotated Factor Pattern (Varimax)

MANOVA
Multivariate Analysis of Variance (MANOVA) was performed to explore the relationship between each of
the independent variables, gender, race, and population density, and the dependent variables, our
extracted factors for academic success (Factor 1), academic environment (Factor 2), and delinquency or
at-risk behaviors (Factor 3). The null hypotheses are: There is no relationship between gender and
Factors 1-3; There is no relationship between ethnicity and Factors 1-3; and there is no relationship
between population density and Factors 1-3.
MANOVA was performed in SASTM Studio using the GLM procedure (generalized linear model). The
PLOTS = ALL option was used to request all applicable plots, but the output was restricted to plots with
less than 5,000 points by the Output Delivery System (ODS) settings. The independent variables were
the categorical variables for gender (V7202), race (V1070) and population density (pop_density). Gender
has two levels, male (1) and female (2). Race has three levels, Black (1), White (2), and Hispanic (3).
Population density has three levels, low-density (0), moderate-density (1), and high-density (2). For
population density, we are primarily interested in the differences between high-density population centers
and all other environments, so a contrast statement will be used to code this comparison. All hypotheses
will be evaluated in the initial analysis. If necessary, post-hoc analysis will be performed by Tukey’s
Studentized Range Test using a Bonferroni correction for multiple testing.

RESULTS
Observations for 17,719 (53.9%) respondents were used to perform multivariate analysis of variance in
SASTM Studio using PROC GLM. The dependent variables were factors for academic achievement,
academic environment, and delinquency/at-risk behavior. A statistically significant overall effect was

found for each independent variable: gender (Wilks’ Lambda = 0.96, F= 273.23, df= (3, 17711), p <
0.0001); race (Wilks’ Lambda = 0.86, F= 479.83, df= (6, 35422), p < 0.0001); population density (Wilks’
Lambda = 0.98, F= 65.31, df= (6, 35422), p < 0.0001). The test statistics table for gender is shown in
Figure 5. The tables for race and population density are given in Figure 14 and Figure 15, Appendix B.
The null hypothesis of no overall effect for each independent variable was rejected. For population
density, the contrast of high-density vs. all others was also statistically significant (Wilks’ Lambda = 0.98,
F= 93.86, df= (3, 17711), p < 0.0001), though the percent of variance explained was even lower (Pillai’s
trace = 0.016, p < 0.0001) than in the analysis including all levels (Pillai’s trace = 0.022, p < 0.0001).
Population density explains approximately 2% of the variance in academic achievement, academic
environment, and at-risk behaviors. Gender explains 4% (Pillai’s trace = 0.044, p < 0.0001) of the
variance, and race explains 14% (Pillai’s trace = 0.145, p < 0.0001) of the variance.

Figure 5. MANOVA Output for the Null Hypothesis for Gender
The Type III Sum of Squares tables for each factor (Figure 16 - Figure 18, Appendix B) indicate that
gender (p < 0.0001) and ethnicity (p < 0.0001) were statistically significantly related to academic
achievement, though population density (p = 0.1506) was not. Ethnicity (p < 0.0001) and population
density (p < 0.0001) were statistically significantly related to academic environment, which gender was
not (p = 0.1150). Only population density (p < 0.0001) was shown to be statistically significantly related to
Factor 3 (delinquency/at-risk behavior). Gender (p = 0.6338) and race (p = 0.5734) were not statistically
significantly related to at-risk behaviors.
Post-hoc analysis by Tukey’s Studentized Range Test was performed using a Bonferroni correction for
multiple testing. Bonferroni correction was performed by dividing the desired level of statistical
significance (α = 0.05) by the number of tests to be performed (21). The corrected alpha of 0.0024 was
specified in a MEANS statement along with the TUKEY and CLDIFF (confidence levels of the estimated
means) options. While the MEANS statement allows us to specify a different alpha level than the
MANOVA analysis, it does not adjust for other terms in the model. The LSMEANS statement should be
used to explore the adjusted means. The post-hoc analysis is summarized in Table 2.
Dep.
Variable

Indep.
Variable

Comparison

Difference
Between
Means

99.76% CI

Sig. *

***

Factor 1

Gender

Female – Male

0.420

0.375, 0.464

Factor 2

Gender

Female – Male

0.022

-0.020, 0.065

Factor 3

Gender

Female – Male

0.006

-0.039, 0.052

Factor 1

Race

White – Black

0.048

-0.027, 0.125

White – Hispanic

0.132

0.072, 0.192

Black – Hispanic

0.084

-0.003, 0.170

White – Black

0.457

0.384, 0.530

***

White – Hispanic

0.853

0.796, 0.910

***

Factor 2

Race

***

Dep.
Variable

Factor 3

Factor 1

Factor 2

Factor 3

Indep.
Variable

Race

Pop_density

Pop_density

Pop_density

Comparison

Difference
Between
Means

99.76% CI

Sig. *

Black – Hispanic

0.396

0.313, 0.479

***

Black – Hispanic

0.017

-0.017, 0.106

White – Black

-0.016

-0.093, 0.062

White – Hispanic

0.002

-0.059, 0.063

High – Moderate

-0.014

-0.071, 0.043

High – Low

0.005

-0.063, 0.073

Moderate – Low

0.019

-0.044, 0.082

High – Moderate

0.064

0.009, 0.118

***

High – Low

0.158

0.091, 0.221

***

Moderate – Low

0.092

0.032, 0.152

***

High – Moderate

0.026

-0.032, 0.084

High – Low

-0.064

-0.134, 0.006

Moderate – Low

-0.090

-0.154, -0.026

***

* Comparisons significant at the α = 0.0024 level are indicated by ***.
Table 2. Post-hoc Analysis Summary.
While statistically significant, the effect of population density was very small across Factors 2 and 3. While
LSMEANS plots for population density and the academic environment factor shows a small, positive
linear trend of increasing factor score with increasing population density, the LSMEANS plot for at-risk
behaviors shows a U-shaped trend (Figure 6) that may be interesting to explore in future analyses.
Please note that the difference between means for high-density and moderate-density was not shown to
be statistically significant.

Figure 6. LSMeans Plot for At-Risk Behaviors (Factor 3) vs Population Density

CONCLUSION
This multivariate analysis of variance explored the relationships between three demographic independent
variables and factors representing academic success, academic environment, and at-risk behaviors. This
analysis has shown that the academic environment for 8th and 10th grade students, as described by
Factor 2—which had high loadings for the variables for parental education, college preparatory program,
and remedial schooling (negative loading)—varies significantly by race. The mean factor score for white
students is highest, while Hispanic students have the lowest mean score for this factor. Population
density explained a very low amount of the variance (2%) and should not be included in future exploratory
or general analyses, as post-hoc analysis is strongly penalized for unnecessary groupings during multiple
testing correction of the significance level.

APPENDIX A: SAS STUDIO CODE
The SAS Studio code used to generate the analyses and figures for this paper is:
libname mydata "/courses/d82c65e5ba27fe300/c_7153/" access=readonly;

proc contents data=mydata.da36799p1;
run;
data fa_data;
set mydata.da36799p1;
keep V7102 V7107 V7215 V7216 V7221 V7222 V7223 V7226 V7231 V7232 V7233
V7253 V7331 V7334 v501 v508 v509 v7202 v1070;
run;
/* recoding and parceling */
data fa_data_parcels;
set fa_data;
* recode for college prep = 1, no = 0;
if v7222 > 1 then
v7222=0;
*recode to move "don't know" to missing (.);
if v7215 > 6 then
v7215=.;
if v7216 > 6 then
v7216=.;
*parent education 7215 + 7216;
parent_ed=v7215 + v7216;
*remedial school;
rem_school=v7232 + v7233;
*population density <- for later MANOVA
per codebook, additive combination of v508 and v509 yield
0 = lowest density, 1 = moderate density, 2 = highest density;
* using '+' to drop NA values, assuming NA = 0 is misleading;
pop_density=v508 + v509;
label parent_ed="Parental education" rem_school="Remedial schooling"
pop_density="Population density";
drop v508 v509 v7215 v7216 v7232 v7233;
run;
title " Descriptive statistics for numeric variables & pop_density";
proc means data=fa_data_parcels n nmiss std min mean median max;
var V7102 V7107 V7221 V7222 V7223 V7226 V7231 V7253 V7331 V7334 parent_ed
rem_school pop_density;
*demographics removed except pop_density;
run;
title;
proc univariate data=fa_data_parcels;
var V7102 V7107 V7221 V7222 V7223 V7226 V7231 V7253 V7331 V7334 parent_ed
rem_school;
histogram / normal kernel;
run;
proc freq data=fa_data_parcels;
tables v501 v7202 v1070 pop_density;
run;
/* Missing data check */
Data check;
set fa_data_parcels;
array chckmiss{*} V7102 V7107 V7221 V7222 V7223 V7226 V7231 V7253 V7331

V7334 parent_ed rem_school v7202 v1070 pop_density;
missdata=0;
do i=1 to dim(chckmiss);
if chckmiss{i}=. then missdata=missdata + 1;
end;
if missdata > 0 then
anymiss=1;
else
anymiss=0;
run;
/*Check for missing data differences by grade*/
proc freq data=check;
tables anymiss missdata;
run;
proc freq data=check;
tables anymiss*(v501 v7202 v1070) / chisq;
run;
proc ttest data=check;
class v501;
var missdata;
run;
/* Using settings from best FA from exploratory analysis */
Title " Factor Analysis: PCA varimax, parcels";
proc factor data=fa_data_parcels rotate=varimax reorder scree
nfactors=3 out=fa_scored;
var V7102 V7107 V7221 V7222 V7223 V7226 V7231 V7253 V7331 V7334 parent_ed
rem_school;
run;
Data manova_data;
set fa_scored;
if factor1 ne .;
label factor1="achievement" factor2="environment" factor3="at-risk
behavior";
run;
/*Are the DVs correlated? */
proc corr data=manova_data;
var factor1 - factor3;
run;
/* MANOVA factors1-3 by gender, race and pop density*/
proc glm data=manova_data plots=all;
class v7202 v1070 pop_density;
model factor1 - factor3=v7202 v1070 pop_density;
CONTRAST 'High-density vs rest' pop_density 1 1 -2;
Manova h=_all_ / printe printh;
lsmeans v7202 v1070 pop_density;
means v7202 v1070 pop_density/ alpha=0.0024 cldiff tukey;
run;

APPENDIX B: SUPPLEMENTAL FIGURES

Figure 7. Missing Data Evaluation

Figure 8. Missing Values by Grade Level

Figure 9. Missing Values by Gender

Figure 10. Missing Values by Race

Figure 11. T-test for Number of Missing Values by Grade Level

Figure 12. Parallel Analysis Criterion Script Output

Figure 13. Factor Analysis: Eigenvalues of Correlation Matrix

Figure 14. MANOVA Output for the Null Hypothesis for Race

Figure 15. MANOVA Output for the Null Hypotheses for Population Density

Figure 16. MANOVA Output Factor 1 (Achievement)

Figure 17. MANOVA Output Factor 2 (Environment)

Figure 18. MANOVA Output Factor 3 (At-Risk Behavior)

